

## Diverse nützliche Excel-Funktionen

Funktionen im Tabellenkalkulationsprogramm Microsoft Excel sind vordefinierte Formeln, die Berechnungen unter Verwendung bestimmter Werte (Argumente) in einer bestimmten Reihenfolge (Syntax) ausführen. Als "Excelianer" habe ich im Laufe der Zeit so einige nützliche Funktionen selbst und aus dem Internet zusammengetragen, welche ich hier darstellen möchte.

Übrigens: Es gibt auch ein Leben hinter Excel. Hier gelangst Du zu meiner [Homepage-Einstiegsseite](#).

Folgende Funktionen stehen zur Auswahl (Stand 03/2005)		
1. Mathematische Funktionen → <a href="#">RUNDEN</a> → <a href="#">GERADE</a> → <a href="#">GANZZAHL</a> → <a href="#">ABRUNDEN</a> → <a href="#">UNTERGRENZE</a> → <a href="#">SUMMEWENN</a> → <a href="#">SUMMENPRODUKT</a> → <a href="#">TEILERGEBNIS</a> → <a href="#">ABS</a>  2. Logische Funktionen → <a href="#">SVERWEIS</a> → <a href="#">VERGLEICH</a> → <a href="#">INDIREKT</a> → <a href="#">INDEX</a> → <a href="#">WAHL</a>	→ <a href="#">ZEILE</a> → <a href="#">BEREICH.VERSCHIEBEN</a>  3. Informationsfunktionen → <a href="#">NV</a> → <a href="#">TYP</a> → <a href="#">ZELLE</a> → <a href="#">ISTFEHLER</a> → <a href="#">VORZEICHEN</a> → <a href="#">ZÄHLENWENN</a>  4. Datumsfunktionen → <a href="#">Einleitung</a> → <a href="#">DATUM</a> → <a href="#">DATEDIF</a> → <a href="#">EDATUM</a> → <a href="#">MONATSENDE</a>	→ <a href="#">KALENDERWOCHE</a> → <a href="#">WOCHENTAG</a>  5. Textfunktionen → <a href="#">TEXT</a> → <a href="#">TEIL</a> → <a href="#">ANZAHL</a> → <a href="#">SUCHEN</a> → <a href="#">GLÄTTEN</a> → <a href="#">VERKETTEN</a> → <a href="#">ERSETZEN</a> → <a href="#">WECHSELN</a> → <a href="#">WIEDERHOLEN</a>  <a href="#">Exkurs: Matrixformeln</a> <a href="#">Exkurs2: Gültigkeitsprüfungen</a>

### 1. Mathematische Funktionen

**Funktion "RUNDEN": rundet eine Zahl *kaufmännisch* auf eine bestimmte Anzahl von Dezimalstellen**

**Syntax:** RUNDEN(Zahl;Anzahl\_Stellen)

- Zahl = zu rundende Zahl
- Anzahl\_Stellen = legt die Anzahl an Dezimalstellen fest

**Beispiel 1:** Die Zahl wird auf die angegebene Anzahl an Dezimalstellen gerundet (hier 1 Dezimalstelle)  
 RUNDEN(2,15;1) ergibt 2,2  
 RUNDEN(2,148;1) ergibt 2,1

**Beispiel 2:** Die Zahl wird auf die nächste ganze Zahl gerundet (hier also ohne Dezimalstellen)  
 RUNDEN(2,5;0) ergibt 3  
 RUNDEN(2,49;0) ergibt 2

**Beispiel 3:** Die Zahl *links* vor dem Dezimalzeichen wird gerundet (also auf ganze Zehner)  
 RUNDEN(44;-1) ergibt 40  
 RUNDEN(45;-1) ergibt 50

**Beispiel 4:** Die Zahl wird auf 5er-Schritte gerundet (das Vielfache, auf das gerundet wird, ist hier die Zahl 5)  
 RUNDEN(13/5;0)\*5 ergibt 15  
 RUNDEN(12/5;0)\*5 ergibt 10  
 (Wichtig sind hierbei die in der Funktion enthaltenen "dividiert durch 5" und "multipliziert mit 5", um zum gewünschten Ergebnis zu kommen)

**Beispiel 5:** Die Zahl wird auf 1000er-Schritte gerundet

RUNDEN(112500/1000;0)\*1000 ergibt 100.000

RUNDEN(64999/1000;0)\*1000 ergibt 65.000

(Wichtig sind hierbei die in der Funktion enthaltenen "dividiert durch 1000" und "multipliziert mit 1000", um zum gewünschten Ergebnis zu kommen)



### Funktion "GERADE": rundet eine Zahl *immer* auf die nächste ganze gerade Zahl kaufmännisch auf oder ab

**Syntax:** GERADE(Zahl) → Zahl = zu rundende Zahl

**Beispiele:**

GERADE(9) ergibt 10

GERADE(8) ergibt 8

Hinweis: Es existiert in Excel auch die entgegengesetzte Funktion UNGERADE.



### Funktion "GANZZAHL": rundet eine Zahl *immer* zur nächsten ganzen Zahl ab

**Syntax:** GANZZAHL(Zahl) → Zahl: eine reelle Zahl, die man zur nächsten ganzen Zahl abrunden möchte

**Beispiele:**

GANZZAHL(6,1) ergibt 6

GANZZAHL(6,9) ergibt 6



### Funktion "ABRUNDEN": rundet eine Zahl *immer* auf die genannte Anzahl von Stellen ab

**Syntax:** ABRUNDEN(Zahl;Anzahl\_Stellen)

- Zahl = zu rundende Zahl
- Anzahl\_Stellen = legt die Anzahl von Dezimalstellen fest

**Beispiele:**

ABRUNDEN(2,888;2) ergibt 2,88

ABRUNDEN(2,888;1) ergibt 2,8

Hinweis: Es existiert in Excel auch die entgegengesetzte Funktion AUFRUNDEN.



### Funktion "UNTERGRENZE": rundet eine Zahl auf das gewünschte Vielfache einer Zahl hin ab

Im Gegensatz zum Vielfach-Runden mittels der Funktion **RUNDEN** wird mit der Funktion **UNTERGRENZE** immer *abgerundet*.

**Syntax:** UNTERGRENZE(Zahl;Schritt)

- Zahl = zu rundene Zahl
- Schritt = bestimmt den Wert, auf den gerundet werden soll

**Beispiele:**

UNTERGRENZE(99;5) ergibt 95 (= Betrag abgerundet auf 5er Intervall)

UNTERGRENZE(99;10) ergibt 90 (= Betrag abgerundet auf 10er Intervall)

UNTERGRENZE(99;50) ergibt 50 (= Betrag abgerundet auf 50er Intervall)

UNTERGRENZE(09:06;"0:15") ergibt 09:00 (= Uhrzeit abgerundet auf Viertelstundenintervall)

Hinweis: Es existiert in Excel auch die entgegengesetzte Funktion OBERGRENZE.



## Funktion "SUMMEWENN": addieren von Zellen, die eine Bedingung erfüllen

Mit der Funktion **SUMMEWENN** werden die Inhalte von Zellen addiert, die mit dem gewünschtem Suchkriterium übereinstimmen. Es können aber auch Zellen gesucht werden, die mit dem Suchkriterium übereinstimmen, um anschließend den Inhalt der *benachbarten* Zellen zu addieren.

**Syntax:** SUMMEWENN(Bereich; Suchkriterien; Summe\_Bereich)

- erforderliche Argumente:
  - Bereich: Bezug auf den Bereich, in dem man mit den Suchkriterien übereinstimmende Einträge sucht (z.B.: A1:A10)
  - Suchkriterien: Zellbezug, Konstante oder Textzeichenfolge (z.B.: A1, 1, ">=50", "Vorsteuer")  
{auch Stellvertreterzeichen sind erlaubt: z.B. "Sch\*", "X?S",..}
- optionales Argument:
  - Summe\_Bereich: Gibt an, welche Zellen man addieren möchte; ohne dies Argument werden automatisch alle Zellen im definierten Bereich addiert, die mit den Suchkriterien übereinstimmen ( z.B. B1:B10)

**Beispiele:**

(1) Addiere in u.g. Tabelle alle Zellen in Bereich B1:B4, deren Inhalt die Zahl 2 ist: SUMMEWENN(B1:B4;2)  
→ Ergebnis: in der Formelzelle stünde eine 4 (= 2 + 2 aus den Zellen B1 und B3)

(2) Suche alle Einträge im Bereich A1:A4 mit einem Inhalt kleiner als "d" und addiere dann die entsprechenden Werte im Bereich von B1 bis B4:  
SUMMEWENN(A1:A4;"<d";B1:B4) → Ergebnis: in der Formelzelle stünde eine 9 (= 2 + 5 + 2)

	A	B
1	a	2
2	b	5
3	c	2
4	d	10

**Beispiel 2:**

Aus den Zellen A1 bis A10 alle Werte addieren, die größer 50 und kleiner 100 sind:

=SUMME(A1:A10)-SUMMEWENN(A1:A10;"<50";A1:A10)-SUMMEWENN(A1:A10;">100";A1:A10).

→ Ergebnis: in der Formelzelle werden nur die gesuchten Werte addiert.

**Beispiel 3 (nur mit Verwendung der Funktion SUMME):**

In Zelle C7 aus dem Bereich C2 bis C5 alle Werte addieren, die in Bereich A2 bis A5 den Text "Soll" und in

Bereich B2 bis B5 das Jahr "1999" enthalten.

Matrixformel (Eingabe mittels Tastenkombination "Strg, Umschalttaste und Returntaste"):

"=SUMME((A2:A5="Soll")\*(B2:B5=1999)\*(C2:C5))"

→ ergibt in der Formelzelle C7 den Wert 5.

Erläuterung: der Ausdruck A2:A5="Soll" gibt Wahrheitswerte zurück, bei Berechnungen werden diese als 0 (FALSCH) bzw. 1 (WAHR) betrachtet. Die obige Formel kann daher wie folgt "aufgelöst" werden:

=SUMME(({ "Soll"; "Haben"; "Soll"; "Soll" } = "Soll") \* ({ 1999; 1999; 2000; 1999 } = 1999) \* ({ 3; 10; 9; 2 } ))

=SUMME({ WAHR; FALSCH; WAHR; WAHR } \* { WAHR; WAHR; FALSCH; WAHR } \* { 3; 10; 9; 2 } )

=SUMME({ 3; 0; 0; 2 } ) = 5

	A	B	C
1	Soll/Haben	Jahr	Betrag
2	Soll	1999	3
3	Haben	1999	10
4	Soll	2000	9
5	Soll	1999	2
6			
7	Summe Soll-Beträge aus 1999:		5

Anmerkung: Die Matrixformel stammt von Ullrich Schwarz, dieser ist ein so genannter MVP ("Most Valuable Professional") für die deutsche Excel-Newsgruppe ("microsoft.public.de.excel") im Microsoft-Newsserver ("msnews.microsoft.com").

Die Funktionsweise von Matrixformeln wird unter [Exkurs: MATRIXFORMELN](#) näher erläutert.



## Funktion "SUMMENPRODUKT": Multiplizieren & addieren

Die Funktion **SUMMENPRODUKT** multipliziert die sich entsprechenden Komponenten (Zellen) der angegebenen Matrizen (Zellbereiche) miteinander und gibt die Summe dieser Produkte zurück.

**Syntax:** SUMMENPRODUKT(Matrix1;Matrix2, Matrix3;...)

- Matrix1; Matrix2;...: mindestens 2 bis maximal 30 Matrizen, deren Elemente zuerst miteinander multipliziert und dann aufaddiert werden. Die Matrizen müssen bezüglich der Zeilen- und Spaltenanzahl identisch sein (ansonsten kommt der Fehlerwert #WERT. Elemente/Zellen ohne numerischen Inhalts werden von der Funktion als Null erkannt).

### Beispiel 1:

Im Beispiel werden in Zelle C3 die Zellbereiche A1:B1, A2:B2 und A3:B3 zuerst multipliziert und dann die 3 Produkte daraus aufaddiert. In der Funktion SUMMENPRODUKT sind demnach 2 Matrizen (Zellbereich A1:A3 und B1:B3) enthalten.

	A	B	C
1	2	2	
2	3	3	
3	4	4	29
4		Formel >	=SUMMENPRODUKT(\$A\$1:\$A\$3;\$B\$1:\$B\$3)
5		Berechnung >	= 2 * 2 + 3 * 3 + 4 * 4

### Beispiele 2:

- SUMMENPRODUKT((REST(ZEILE(A1:A10);2)=0)\*A1:A10) liefert die Summe der Werte jeder geraden Zeile im Bereich A1 bis A10 (also die Summe aus den Zellen A2, A4, A6, A8 und A10).
- SUMMENPRODUKT((REST(ZEILE(A1:A10);2)=1)\*A1:A10) liefert die Summe der Werte jeder ungeraden Zeile im Bereich A1 bis A10 (also die Summe aus den Zellen A1, A3, A5, A7 und A9).
- SUMMENPRODUKT((REST(ZEILE(A1:A20);5)=0)\*A1:A20) liefert die Summe der Werte jeder 5. Zeile im Bereich A1 bis A20 (also die Summe aus den Zellen A5, A10, A15 und A20).

Hinweis: bei der Funktion [Zeile](#) ist im Beispiel 3 ein ähnlicher Fall mit einer [Matrixformel](#) gelöst worden.



## Funktion "TEILERGEBNIS": Ermitteln der Summe der nicht ausgeblendeten Zellen

Die Funktion **TEILERGEBNIS** berechnet im Gegensatz zur Funktion Summe *nicht* die im Zellenbereich angegebenen ausgeblendeten Zeilen bzw. Spalten mit ein.

*Achtung:* dies funktioniert aber nur bei einer gefilterten Liste. Ansonsten werden die ausgeblendeten Zeilen mitaddiert.

**Syntax:** TEILERGEBNIS(Funktion;Bezug1;Bezug2;...)

- Funktion: eine Zahl von 1 bis 11, die einer der vordefinierten Funktionen entspricht, mit der innerhalb einer Liste ein Teilergebnis berechnet werden soll (bspw. entspricht die Ziffer 9 der Funktion SUMME).
- Bezug1, Bezug2 ...: zwischen 1 und 29 weitere Teilbereiche, für die das Teilergebnis berechnet werden soll.

**Beispiel:** TEILERGEBNIS(9;C3:C5) erzeugt ein Teilergebnis der Zellen C3:C5 mit der Funktion SUMME.



## Funktion "ABS": Zellenwert immer positiv darstellen

Die Funktion **ABS** liefert den Absolutwert einer Zahl: dieser ist die Zahl ohne ihr Vorzeichen → die Funktion wandelt negative Zahlen in positive um; positive Zahlen bleiben dagegen positiv.

**Syntax:** ABS(Zahl) → Zahl: beliebige reelle Zahl

**Beispiel:** Positive und negative Zahlen getrennt summieren.

Sollen Soll- und Habenbeträge aus einer Bilanz getrennt aufsummiert werden, so kann man folgende Matrixformeln verwenden (die Formeln muss man mittels Tastenkombination "Strg-, Hoch- und Return-taste" bestätigen; die geschweiften Klammern kann man nicht selbst eingeben !).

Hinweis: Die Funktionsweise von Matrixformeln wird unter [Exkurs: MATRIXFORMELN](#) näher erläutert.

- Positive Zahlen: {=SUMME(<Bereich>+ABS(<Bereich>))/2}
- Negative Zahlen: {=SUMME(<Bereich>-ABS(<Bereich>))/2}

Die Formel für Errechnung der positiven Zahlen arbeitet folgendermaßen: in der Summenformel werden im ersten Teil alle Zahlen *inklusive* ihrer Vorzeichen addiert. Im zweiten Teil wird über die ABS-Funktion die Summe aller Beträge des Bereiches dazugezählt – *unabhängig* von dessen Vorzeichen.

Die negativen Zahlen werden im ersten Teil abgezogen, da sie mit Vorzeichen summiert werden. Im zweiten Teil werden sie aber wieder zu der Summe dazugezählt, da dort der Betrag addiert wird. Also werden sie einmal abgezogen und einmal dazugezählt – und treten somit in der Summe nicht auf.

Die positiven Zahlen werden dagegen zweimal addiert, so dass das Ergebnis der Summe-Formel am Ende noch durch zwei geteilt werden muss.

Bei der Formel für die Errechnung der negativen Zahlen wird durch das negative Vorzeichen vor der ABS-Funktion die Summierung umgekehrt.

	A		A
1	+ 50		+ 50
2	- 50		- 50
3	- 50		- 50
4	+ 50		+ 50
5	+ 50		+ 50
6	+ 150	<i>Formeln &gt;</i>	{=SUMME((A1:A5)+ABS(A1:A5))/2}
7	- 100	<i>&lt; Darstell.</i>	{=SUMME((A1:A5)-ABS(A1:A5))/2}



## 2. Logische Funktionen

### Funktion "SVERWEIS": Bestimmte Vergleichswerte in einer anderen Tabelle suchen

Die Funktion **SVERWEIS** prüft, ob die (*anzugebende*) Spalte einer Matrix / Tabelle einen bestimmten Wert enthält, und liefert dann in einer anderen Zelle einen Wert, der in einer weiteren, anzugebenden Zelle steht. Die Funktion sollte man immer dann verwenden, wenn Vergleichswerte in einer Spalte stehen, die sich rechts (!) von den Daten befinden, nach denen man sucht.

**Syntax:** SVERWEIS(Suchkriterium; Matrix; Spaltenindex; Bereich\_Verweis)

- Suchkriterium: Der Wert, nach dem man in der ersten Spalte der Matrix sucht (dieses Suchkriterium kann ein Wert, ein Bezug oder eine Zeichenfolge [Text] sein)
- Matrix: Die Informationstabelle, in der Daten gesucht werden (am besten verwendet man einen Bereich oder Bereichsnamen, wie z.B. "Datenbank" oder "Liste")
- Spaltenindex: Die Nr. der Spalte in der Mehrfachoperationsmatrix, aus der der übereinstimmende Wert geliefert werden soll (ein Spaltenindex von 1 liefert den Wert, der in der ersten Spalte der Matrix steht; ein Spaltenindex von 2 liefert den Wert, der in der zweiten Spalte der Matrix steht usw.)
- Bereich\_Verweis: ein Wahrheitswert, der angibt, wie die zur Matrix gehörenden Werte sortiert sind. Ist Bereich\_Verweis gleich WAHR oder nicht angegeben, müssen die Werte, die die erste Spalte der Matrix bilden, in aufsteigender Reihenfolge vorliegen; ist Bereich\_Verweis gleich FALSCH, so spielt es keine Rolle, in welcher Reihenfolge die zur Matrix gehörenden Werte angeordnet sind
  - Kann die Funktion SVERWEIS den als Suchkriterium angegebenen Wert nicht finden, und ist Bereich\_Verweis = WAHR, so liefert sie bezüglich dem Suchkriterium den nächst kleineren Wert.
  - Kann die Funktion SVERWEIS den als Suchkriterium angegebenen Wert nicht finden, und ist Bereich\_Verweis gleich FALSCH, liefert SVERWEIS den Fehlerwert #NV.

Anmerkungen:

Die analoge Funktion **WVERWEIS** geht im Gegensatz zu SVERWEIS von einer zeilen- statt spaltenweisen Anordnung der Tabellen aus. Durchsucht wird die erste Zeile eines Bereiches und liefert den zugehörigen Wert in der angegebenen Bereichsspalte.

Der Nachteil an SVERWEIS ist, dass nur Ergebnisse aus Spalten gefunden werden, die sich in einer Spalte *rechts* der Suchspalte befinden. Bei der Funktion **VERWEIS** dagegen kann die Ergebnisspalte auch *links* der Suchspalte sein. Allerdings müssen bei VERWEIS die Inhalte der Suchspalte zwingend in aufsteigender Reihenfolge sein.

**Beispiel:** Durch die SVERWEIS-Funktion in Zelle B2 und B3 der ersten Tabelle (Tab1) sucht Excel in der zweiten Tabelle (Tab2) nach den Bezeichnungen für die betreffende Kostenstelle (hier Kst. 300 und 310) und stellt sie in den entsprechenden Funktions-Zellen dar.

Tab1	A	B		A	B
1	Kostenstelle	Bezeichnung		Kostenstelle	Bezeichnung
2	300	Offsetdruck	< Darstellung	300	=SVERWEIS((A2);Tab2!\$A\$2:\$B\$7;2,FALSCH)
3	310	Buchdruck	Formeln >	310	=SVERWEIS((A3);Tab2!\$A\$2:\$B\$7;2,FALSCH)

Tab2	A	B
1	Kostenstelle	Bezeichnung
2	300	Offsetdruck
3	305	Scheckdruck
4	310	Buchdruck
5	320	Tiefdruck
6	350	Versand
7	390	Sonstiges



## Funktion "VERGLEICH": Die Position eines Elementes innerhalb einer Matrix suchen

Die Funktion **VERGLEICH** liefert als Ergebnis die Zellposition, die der jeweils gefundene Wert innerhalb eine Suchmatrix / Tabelle einnimmt (nicht den Wert selbst - dies geht stattdessen mit der Funktion **SVERWEIS**). Findet VERGLEICH keine übereinstimmenden Werte, so liefert die Funktion den Fehlerwert #NV.

**Syntax:** VERGLEICH(Suchkriterium;Suchmatrix;Vergleichstyp)

- Suchkriterium: der Wert, gegen den man bestimmte Elemente der Suchmatrix abgleichen will. Dies kann ein Wert (Zahl, Text, Wahrheitswert) oder ein Zellbezug zu einem solchen Wert sein. Die Funktion unterscheidet beim Abgleichen von Zeichenfolgen (Text) nicht nach Groß- und Kleinbuchstaben.
- Suchmatrix: ein zusammenhängender Zellbereich mit möglichen Vergleichskriterien.
- Vergleichstyp: gibt an, wie verglichen werden soll. Fehlt das Argument, so wird eine 1 angenommen.
  - bei Vergleichstyp 1 liefert VERGLEICH den größten Wert, der kleiner gleich Suchkriterium ist. Die Elemente der Suchmatrix müssen aufsteigend sortiert sein.
  - bei Vergleichstyp 0 liefert VERGLEICH den ersten Wert, der gleich dem Suchkriterium ist. Die Elementreihenfolge ist beliebig. Bei diesem Vergleichstyp können als Suchkriterium auch Stellvertreterzeichen verwendet werden (Ein Sternchen für eine beliebige Zeichenfolge sowie ein Fragezeichen für ein einzelnes Zeichen).
  - bei Vergleichstyp -1 liefert VERGLEICH den kleinsten Wert, der größer gleich Suchkriterium ist. Die Elemente der Suchmatrix müssen absteigend sortiert sein.

**Beispiel 1:** Steht in Zelle A1 der Text "Zweiter", so liefert die Matrixformel "=VERGLEICH(A1; {"Erster";"Zweiter";"Dritter";"Vierter"};0)" den Wert 2, da dies die relative Position ist, die das Element "Zweiter" in der Matrix {"Erster";"Zweiter";"Dritter";"Vierter"} einnimmt.

**Beispiel 2:** In einer mengenabhängigen Preistabelle einen Artikelpreis suchen. In Zelle B9 sucht die Funktion VERGLEICH in Verbindung mit Funktion INDEX nach einem bestimmten Artikelpreis. Die Vorgaben hierzu sind in Zelle B7 (Artikelmenge) und in Zelle B8 (Artikelname) zu setzen. Je nachdem, welche Anzahl und welcher Artikel in die Zellen geschrieben werden, wird in Zelle B9 der zugehörige Artikelpreis ausgewiesen.

	A	B	C	D	E	F
1	Artikel \ Anzahl	1	2	3	5	10
2	Schrauben	0,50 €	0,90 €	2,00 €	3,90 €	35,00 €
3	Muttern	0,20 €	0,53 €	0,90 €	1,80 €	17,00 €
4	Nägeln	0,10 €	0,18 €	0,40 €	0,75 €	7,00 €
5	Ösen	0,40 €	0,70 €	1,70 €	3,00 €	25,00 €
6						
7	Anzahl:	3				
8	Artikel:	Muttern				
9	Preis:	0,90 €				
10	Formel in Zelle B9:	=INDEX(B2:F5;VERGLEICH(B8;A2:A5;0);VERGLEICH(B7;B1:F1;0))				
11						

Hinweis: Die Funktionsweise der logischen Funktion INDEX wird unter **INDEX** näher erläutert.



## Funktion "INDIREKT": Den Bezug auf eine in einer Formel befindlichen Zelle ändern, ohne die Formel selbst zu ändern

Die Funktion **INDIREKT** liefert den Bezug eines Textwertes. Diese Bezüge werden sofort ausgewertet, so dass die zu ihnen gehörenden Werte direkt angezeigt werden.

**Syntax:** INDIREKT(Bezug;A1)

- Bezug: Bezug auf eine Zelle, diese muss in Anführungsstrichen gesetzt sein (verweist *Bezug* auf eine

andere Arbeitsmappe, so muss diese geöffnet sein)

- A1: Wahrheitswert, der angibt, welche Art von Bezug in der Zelle enthalten ist; optionales Argument (ist A1 = WAHR oder nicht angegeben, so wird *Bezug* als Bezug in der A1-Schreibweise interpretiert; ist A1 = FALSCH dagegen in der Z1S1-Schreibweise)

**Beispiel 1:** In Formeln den Fehlerwert #BEZUG! (beispielsweise durch Löschen oder Verschieben von Zellen ausgelöst) vermeiden.

Die Funktion INDIREKT setzt beispielsweise - als Summenformel verwendet - beim Löschen einer Spalte automatisch die rechte Spalte daneben als neue Berechnungsgrundlage ein. Verwendet man in Zelle A6 die Formel "=SUMME(INDIREKT("A1:A5"))", so summiert die Funktion die Werte der Zahlen in dem genannten Zellenbereich. Verschiebt man nun den Bereich B1:B5 auf A1:A5, so würde normalerweise die Summenformel den Fehlerwert #BEZUG! angeben, da die alten Zellen überschrieben worden sind. Mittels der Funktion INDIREKT wird jedoch der Bezug aus Spalte B automatisch bei allen Berechnungen an die Stelle der alten Bezüge gesetzt.

**Beispiel 2:** Mittels der Funktion INDIREKT kann ein Zellenbezug aus mehreren Zelleninhalten zusammengesetzt werden.

Verwendet man in Zelle C1 die Formel "=SUMME(INDIREKT("B1:B"&A1))", und befindet sich in Zelle A1 der Wert 5, so summiert die Funktion die Zelleninhalte aus dem Bereich B1 bis B5.

Erklärung: B1 ist in der Formel vorgegeben, der zweite Zellbezug (B5) wurde durch das ebenfalls vorgegebene B plus den Inhalt aus Zelle A1 ermittelt: nämlich der Zahl 5. Der zweite Zellenbezug wurde also aus 2 verschiedenen Teilen variabel zusammengesetzt.



## Funktion "INDEX": Den Inhalt einer Zelle mittels Positionsangaben abfragen

In der Excel-Hilfe steht hierzu folgendes: *"Die Funktion INDEX liefert den Wert eines Elementes in einer Tabelle oder einer Matrix, das innerhalb der Matrix die Position einnimmt, die durch für die Zeile sowie die Spalte angegebenen Indizes festgelegt ist"*. Zugegeben, hört sich recht kompliziert an. Einfacher gesagt, mit der Funktion ermittelt man den Zelleninhalt des Schnittpunktes 2er Positionsangaben (nämlich der Zeilen- und Spaltennummer) eines Tabellenbereiches.

**Syntax:** INDEX(<Matrix>;<Zeile>;<Spalte>)

- Matrix: der zu durchsuchende Zellbereich (Tabelle)
- Zeile: die Nummer der gesuchten Zeile als relative Bezugszahl (1. Positionsangabe)
- Spalte: die Nummer der gesuchten Spalte als relative Bezugszahl (2. Positionsangabe)

**Beispiel 1:** Enthalten die Zellen D1:D2 den Text Äpfel und Bananen und die Zellen E1:E2 den Text Zitronen und Pfirsiche, dann ermittelt die Formel "=INDEX(D1:E2;2;2)" den Textinhalt "Pfirsiche" aus Zelle E2. Der relative Bezug der gefundenen Zelle ist nämlich die 2. Zeile und 2. Spalte des Tabellenbereiches D1 bis E2.

**Beispiel 2:** Bestimmte Werte, abhängig von einem anderen Wert, summieren.

In Zelle A1 steht die Zahl 5. In den Zellen C1 bis C10 stehen untereinander die Werte 1, 2, 3 u.s.w. bis 10. Nun sollen in Spalte C alle Zellen addiert werden, bis in den Zelleninhalten der Wert aus Zelle A1 - nämlich 5 - erreicht wird

→ Excel addiert in der Formel 1, 2, 3, 4 und 5 - ergibt also den Wert 15.

Formel: "=SUMME(C1:INDEX(C1:C10;A1))".

Erläuterung: Mit Hilfe der Funktion INDEX wird der Wert der Zelle, deren Position durch den Eintrag in A1 gekennzeichnet wurde, berechnet. Die Funktion SUMME errechnet danach die Summe der Zelleninhalte von C1 bis zu der Zelle, welche durch die Funktion INDEX ermittelt wurde (im Beispiel die Zelle C5 mit dem gesuchten Wert 5).

**Nachfolgende INDEX-Funktionen ermitteln die unterste Zelle mit einer Zahl als Inhalt und stellen diese in der Formelzelle dar. Möglich ist aber auch die Ermittlung von Textinhalten.**

### 5a) Möglichkeit Nr. 1: falls keine leeren Zellen im zu durchsuchenden Bereich existieren.

Die Funktion **INDEX** in Verbindung mit **ANZAHL2** ermittelt den letzten Wert bzw. Text, der in einer Zeile oder Spalte steht. Voraussetzung ist, dass keine Leerzellen dazwischen vorkommen.

**Syntax:** INDEX(<Bereich>;ANZAHL2(<Bereich>))

**Beispiel 1:** Formel für die letzte Zelle mit Inhalt aus Zeile 1: "=INDEX(1:1;ANZAHL2(1:1))".

**Beispiel 2:** Formel für die letzte Zelle mit Inhalt aus Spalte A: "=INDEX(A:A;ANZAHL2(A:A))".

### 5b) Möglichkeit Nr. 2: falls Leerzellen im zu durchsuchenden Bereich existieren.

Die Funktionen **INDEX**, **VERGLEICH**, **WENN** und **ISTLEER** ermitteln innerhalb u.g. Matrixformel den gesuchten Zelleninhalt. Eingabe der Matrixformel: mittels geschweiffter Klammern, die man aber nicht selbst eingeben darf, sondern nur über die Tastenkombination "Strg, Umschalttaste und Return" aufruft.

Anmerkung: Die Funktionsweise von Matrixformeln wird unter **Exkurs: MATRIXFORMELN** näher erläutert.

**Syntax:** {=INDEX(<Bereich>;VERGLEICH(WAHR;WENN(ISTLEER(<Bereich>);<Bereich>);1);0)}

- Bereich: man gibt den Bezug auf den Zellbereich an, aus dem man den Inhalt der untersten nicht leeren Zelle ermitteln möchte (z.B. A2:A6).

**Beispiel:** Es werden Leerzellen dazwischen erkannt und die unterste Inhaltzelle ermittelt (hier die Zahl 8).

	A	B		A	B
1	Be- reich	Ergeb- nis		Be- reich	Ergebnis
2	2	8	Formeln >	2	{=INDEX(A2:A7;VERGLEICH(WAHR;WENN(ISTLEER(A2:A7);A2:A7);1);0)}
3	5		< Darstell.	5	
4					
5	5			5	
6	8			8	



## Funktion "WAHL": Auswählen eines Zelleninhaltes aus bis zu 29 Inhalten anhand einer Indezzahl

Sind beispielsweise die Argumente Wert 1 bis Wert 7 mit den Wochentagsnamen belegt, liefert **WAHL** einen dieser Tage, wenn eine der Zahlen 1 bis 7 als Index angegeben wird.

Hinweis: Im Vergleich mit der **WENN**-Funktion können mit WENN nur maximal 7 Verschachtelungen (d.h. Wahlmöglichkeiten) realisiert werden. WAHL bietet dagegen 29 Wahlmöglichkeiten (siehe hierzu Beispiel 2)

**Syntax:** WAHL(Index; Wert1; Wert2; Wert3; ... )

- Index: gibt an, welcher Argumentwert gewählt ist  
→ Index darf jeden Wert von 1 bis 29 annehmen; ansonsten ergibt sich der Fehlerwert "Wert" (ist Index gleich 1, liefert WAHL den Wert1; ist Index gleich 2, liefert WAHL den Wert2 usw.).  
→ ist Index keine ganze Zahl, so werden die Nachkommastellen abgeschnitten.
- Wert: kann direkt als Zahl, Text oder indirekt als Formel bzw. Bezug auf eine Zelle angegeben werden.

**Beispiel 1:** Enthält die Zelle A10 die Zahl 3, dann liefert WAHL(A10;"Erster";"Zweiter";"Dritter";"Vierter") das Ergebnis "Dritter".

**Beispiel 2:** Die geplante Produktionsmenge eines Monats anhand des aktuellen Tagesdatums ermitteln.

Mit der WENN-Funktion ließe sich das Beispiel nicht lösen, da die Funktion maximal 7 Verschachtelungen zulässt, hier aber insgesamt 12 Möglichkeiten (Monate) zu berücksichtigen sind. Mit Hilfe der WAHL-Funktion bieten sich dagegen bis zu 29 Wahlmöglichkeiten.

In der Tabelle sollen die geplanten Produktionsmengen aus jedem Monat anhand des aktuellen Tagesdatums heraus gelesen werden. Ist das Datum in Zelle A5 der 10.10.2000, so ermittelt die MONAT-Funktion die Zahl 10 (für Oktober), daraufhin selektiert die WAHL-Funktion den 10. Wert aus seiner Wertaufzählung und stellt sie in Zelle B5 dar.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Monat:	Jan	Feb	Mrz	Apr	Mai	Jun	Jul	Aug	Sep	Okt	Nov	Dez
2	Anzahl (gepl.)	100	100	150	150	200	250	200	150	150	150	100	100
3													
4	Datum												
5	10.10.00	150	geplante Menge im aktuellen Monat										
6	=HEUTE()	=WAHL(MONAT(A5);B2;C2;D2;E2;F2;G2;H2;I2;J2;K2;L2;M2)											



## Funktion "ZEILE": Liefert die Zeilennummer eines Bezuges

**Syntax:** ZEILE(Bezug) → Bezug ist die Zelle oder der Zellbereich, deren Zeilennummer(n) man ermitteln will [nur optionales Argument]

**Beispiel 1:** Nummerieren von Datensätzen.

Wenn in einer Tabelle mit aufeinander folgenden Datensätzen (von Spalte B bis Spaltenende; in Zeile 1 sind die Spaltenüberschriften) ab Zeile 2 in der Spalte A abwärts alle vorhandenen Datensätze durchnummeriert werden sollen, so muss man die Formel "=ZEILE(A2)-1" in Zelle A2 erstellen und hiernach die Formel (mit relativem Zellenbezug) bis Datensatzende kopieren.

Vorteil: Werden später einmal Zwischenzeilen mit neuen Datensätzen eingefügt, so wird die Listennummerierung automatisch angepasst.

Sollte oberhalb der Liste eine zusätzliche Zeile für weitere Überschriften eingefügt werden, so ändert man die Formel auf "=ZEILE(A3)-ZEILE(\$A\$2)" und kopiert sie abwärts. Hierdurch werden die nummerierten Zeilen nur ab der 3. Zeile angepasst (ohne den absoluten Zellbezug hätte die Nummerierung der Zellen bei Nummer 2 begonnen anstatt auf 1). Das spätere Einfügen weiterer Zeilen oberhalb des eigentlichen Listbereiches hat nun keine negativen Auswirkungen mehr auf die darunter liegende Zeilennummerierung.

**Beispiel 2:** Die Zeilennummer wird in die Formelzelle (hier Spalte A) dargestellt, wenn in der Zelle rechts daneben ein Eintrag vorhanden ist. Zur besseren Darstellung formatiert man die Zellen mit dem Zahlenformat "Standard".

	A	B		A	B
1	Zeile 1	Hallo		=WENN(NICHT(ISTLEER(B1));ZEILE();"")	Hallo
2			<i>Formeln &gt;</i>	=WENN(NICHT(ISTLEER(B2));ZEILE();"")	
3	Zeile 3	115	<i>&lt; Darstell.</i>	=WENN(NICHT(ISTLEER(B3));ZEILE();"")	115

**Beispiel 3:** Jede x-te Zelle summieren.

Im Bereich A1:A10 soll jede zweite Zelle addiert werden (also die Zellen A2, A4, A6, A8 und A10).  
Matrixformel: "=SUMME(WENN(REST(ZEILE(A1:A10);2)=0;A1:A10))". Die Formel muss man mittels Tastenkombination "Strg-, Hoch- und Return Taste" bestätigen; die geschweiften Klammern kann man nicht selbst eingeben.

Hinweis 1: Die Funktionsweise von Matrixformeln wird unter [Exkurs: MATRIXFORMELN](#) näher erläutert.

Hinweis 2: Unter Verwendung der Funktion [SUMMENPRODUKT](#) ist ein ähnlicher Beispielfall ohne Matrixformel gelöst worden.

**Beispiel 4:** Mittels bedingter Formatierung bei geraden Zeilennummern den Zellenhintergrund farbig formatieren.

Im Bereich A1:A20 soll jede zweite Zelle farblich gelb hinterlegt werden (also die Zellen A2, A4, A6, usw.).  
Vorgehensweise: den Zellbereich A1 bis A20 markieren, per Menü *Format* und Untermenü *Bedingte Formatierung...* beim Bereich "Formel ist" die Option "Bedingung 1" auswählen. Dann rechts daneben die

Formel "`=REST(ZEILE();2)=0`" eintragen. Zum Schluss das gewünschte Format mittels Befehl *Format...* auswählen.



## Funktion "BEREICH.VERSCHIEBEN": Liefert einen Bezug, der gegenüber dem angegebenen Bezug versetzt ist

Die Funktion **BEREICH.VERSCHIEBEN** verschiebt weder Zellen, noch ändert sie die Markierung (Auswahl); sie liefert nur einen gegenüber dem Ausgangsbezug versetzten Bezug, der in Berechnungen verwendet werden kann. Selbständig als Formel findet sie daher keine Verwendung.

**Syntax:** BEREICH.VERSCHIEBEN(Bezug; Zeilen, Spalten, Höhe, Breite)

- **Bezug:** Der Zellbezug, der als Ausgangspunkt der Verschiebung dienen soll.
- **Zeilen:** Anzahl an Zeilen, um die die obere linke Eckzelle des Bereiches nach oben (negative Anzahl) oder unten (positive Anzahl) verschoben werden soll. Ist das Argument Zeilen = 5, so liegt die obere linke Ecke des neuen Bezugs 5 Zeilen unterhalb des Ausgangsbezugs (siehe Argument *Bezug*).
- **Spalten:** Anzahl an Spalten, um die die obere linke Eckzelle des Bereiches nach links (negative Anzahl) oder rechts (positive Anzahl) verschoben werden soll. Ist das Argument Spalten = 5, so liegt die obere linke Ecke des neuen Bezugs 5 Spalten rechts des Ausgangsbezugs (siehe Argument *Bezug*).
- **Höhe:** Höhe des neuen Bezugs in Zeilen (die Zahl muss positiv sein); optionales Argument
- **Breite:** Breite des neuen Bezugs in Spalten (die Zahl muss positiv sein); optionales Argument

**Beispiel 1:** "`=SUMME(BEREICH.VERSCHIEBEN(C2;1;2;3;1))`" berechnet die Summe eines aus 3 Zeilen und 1 Spalte umfassenden Bereichs, der sich 1 Zeile unterhalb und 2 Spalten rechts der Zelle C2 befindet (= Zellbezug E3 bis E5).

**Beispiel 2:** Im Bereich A1:A20 stehen 20 (maximal) aufzusummierenden Werte. In Zelle C1 steht ein Wert, nach welchem in Zelle C2 so viele Werte aus Spalte A aufsummiert werden sollen, wie in Zelle C1 als Anzahl angegeben sind.

Formel in Zelle C2: "`=SUMME(BEREICH.VERSCHIEBEN(A1:A20;0;0;C1;1))`". Steht z.B. in Zelle C1 die Zahl 7, so werden in Zelle C2 die Zellen A1 bis A7 aufsummiert, der Rest bleibt unberücksichtigt.

**Beispiel 3:** In Zelle K4 soll, abhängig vom in Zelle D4 als Zahl anzugebendem Monat, die kumulierten Umsatzzahlen geliefert werden. Gibt man in Zelle D4 die Zahl 5 für den Monat Mai an, so rechnet die Formel die Umsatzzahlen von Monat Januar bis Mai (= Zelle B2 bis F2) zusammen.

Die Syntax der Formel ersieht man in der Tabelle.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Monat:	Jan	Feb	Mrz	Apr	Mai	Jun	Jul	Aug	Sep	Okt	Nov	Dez
2	Umsatz:	50	55	60	65	70	70	80	85	70	60	50	40
3													
4	Bestimme Monat:			5							300		
	Formel in Zelle K4: <code>=SUMME(BEREICH.VERSCHIEBEN(\$B\$2;0;0;1;\$D\$4))</code>												

**Beispiel 4:** Der Umsatz von 1999 in Spalte B soll mit dem Umsatz von 2000 in Spalte C verglichen werden. In der Spalte für 1999 sind alle Monate eingetragen, da es das Vorjahr betrifft. In der Spalte C für das aktuelle Jahr 2000 dagegen nur die bis zum laufenden Monat.

Die Formel in Zelle B15 ermittelt aus dem Jahr 1999 nur die Werte, bei der die kumulierten Werte aus der 2000er Spalte mit denen aus 1999 bis zum gleichen Monat vergleichbar sind. Im Beispiel werden also nur die Zahlen von Januar bis Juni 1999 aufsummiert (= Zellbereich B2 bis B7), um sie mit dem aktuellen Stand im Juni 2000 vergleichen zu können.

Die Syntax der Formel ersieht man in der Tabelle.

	A	B	C	D
1	Monat / Jahr	GJ 1999	GJ 2000	
2	Januar	100	90	
3	Februar	105	120	
4	März	110	100	
5	April	115	120	
6	Mai	120	140	
7	Juni	130	130	
8	Juli	120		
9	August	115		
10	September	110		
11	Oktober	100		
12	November	90		
13	Dezember	80		
14		Summe 1999	Summe 2000	Differenz
15		680	700	+ 20
	Formeln in		=SUMME(C2:C13)	=C15-B15
	Zeile 15	=SUMME(BEREICH.VERSCHIEBEN(B2;0;0;ANZAHL(C2:C13)))		

Hinweis: Die Funktionsweise der Textfunktion ANZAHL wird unter [ANZAHL](#) näher erläutert.



## 2. Informationsfunktionen

### Funktion "NV": Liefert den Fehlerwert "#NV"

Die Funktion **NV** (bzw. dessen Fehlerwert) besagt, dass kein Wert verfügbar ist. Man verhindert damit, dass die Zellen, die diese Funktion enthalten, in Formelberechnungen unbeabsichtigt einbezogen werden.

**Syntax:** NV() → zusammen mit dem Namen der Funktion muss ein leeres Klammerpaar angegeben werden (Hinweis: alternativ kann #NV verwendet werden).

**Beispiel 1:** Enthalten die Zellen A1 bis A4 die Inhalte 1, 0, #NV und 4, so ermittelt die Funktion =ANZAHL(A1:A4) die Zahl 3, da nur drei Wertinhalte von der Funktion entdeckt wurden. Die Zelle mit #NV wurde dagegen nicht erkannt (statt #NV kann auch =NV() verwendet werden).

**Beispiel 2:** In Liniendiagrammen Nullwerte unterdrücken.

In einem Liniendiagramm sollen aktuelle Monatsumsätze mit den Umsätzen aus dem vergangenen Geschäftsjahr verglichen werden. Innerhalb des aktuellen Jahres hat man nun das Problem, das noch bevorstehende Monate als Nullwerte oder leere Zellen in der Umsatztable enthalten sind. Im dazugehörigen Diagramm wird ab einem solchen Monat die Diagrammlinie natürlich auf den Wert Null gesenkt (i.d.R. auf die X-Achse). Mit #NV verhindert man, dass diese Linie von Excel überhaupt gezogen wird.

Man verwendet oberhalb der Zeile mit den aktuellen Monatsumsätzen (diese befinden sich bspw. in Zeile 4) diese Zeile als Hilfszeile (Zeile 3). In Zelle A3 wird folgende Formel eingebaut: =WENN(A4=0;NV();A4). Nachfolgende Formelzellen werden auf die Zellen B4 bis L4 verwiesen. Auf diese - ausblendbare - Hilfszeile muss das Diagramm wertemäßig bezogen werden. Nullwerte werden nun im Diagramm unterdrückt.



### Funktion "TYP": Liefert den "Typ von Wert"

Die Funktion **TYP** kann man immer dann einsetzen, wenn die weitere Vorgehensweise einer Funktion davon abhängt, welchen Typ der in einer bestimmten Zelle enthaltene Wert hat.

**Syntax:** TYP(Wert) → Wert: kann ein beliebiger Excel-Wert sein (bspw. eine Zahl, ein Text, ein Wahrheitswert u.s.w.). Ist *Wert* eine Zahl, so liefert *TYP* eine 1 (bei einer Formel dagegen die 8, einem Text die 2, einem Fehlerwert die 16, einem Wahrheitswert die 4 und bei einer Matrix die 64)

**Beispiel 1:** Enthält A1 als Zeichenfolge den Text "Schmidt", dann folgt TYP(A1) = 2.

**Beispiel 2:** Enthält die Zelle B1 die Formel "=WENN(TYP(A1)=2;"Richtig !";WENN(TYP(A1)=1;"Falsche Eingabe !";))" und die Zelle A1 einen Textinhalt, so gibt die Formel die Zeichenfolge "Richtig !" zurück. Im Falle einer Zahl in in Zelle A1 dagegen die Zeichenfolge "Falsche Eingabe !".

Hinweis: Mittels einer *Gültigkeitsprüfung* lässt sich bequemer als mit der hier dargestellten WENN-Formel festlegen, welche Daten für einzelne Zellen oder ganze Zellbereiche zulässig sind. Man kann die einzugebenden Daten auf einen bestimmten Datentyp beschränken. - bspw. ganze Zahlen, Texte oder Wahrheitswerte - und Grenzwerte für die gültigen Werte festlegen. Siehe hierzu [Gültigkeitsprüfung](#).



## Funktion "ZELLE": Dateinamen, Blattnamen und Pfad innerhalb einer Zelle anzeigen

Die Funktion **ZELLE** liefert Informationen zu der Formatierung, der Position oder dem Inhalt einer Zelle. In den Beispielen werden der aktuelle Arbeitsmappen-Namen, der Blattnamen und dessen Pfad mittels der Funktion angezeigt. Diese Formel liefert erst dann das gewünschte Ergebnis, wenn die Datei gespeichert wurde.

**Syntax:** ZELLE(Infotyp;Bezug)

- Infotyp: ein Textwert, mit dem man die gewünschte Art von Zellinformationen festlegt (In der Excel-Hilfe findet man eine Liste möglicher Werte für Infotyp und die zugehörigen Ausgabewerte der Funktion).
- Bezug: die Zelle, über die man Informationen wünscht (optionales Argument). Die Angabe einer Zelle - z.B. "A1" - ist dann zweckmäßig, wenn man in anderen Tabellenblätter Veränderungen vorgenommen. Kehrt man danach ins Ursprungsblatt mit der Zelle-Funktion darin zurück. so würde die Formel den falschen Blattnamen liefern. Mit "A1" kann man dagegen die Aktualisierung der Formel erzwingen.

**Beispiele:** Arbeitsmappe "STARTTAB.XLS" mit dem Excel-Blatt "Tabelle1" ist im Verzeichnis "C:\EXCEL" gespeichert

(1) Anzeige des vollständigen Dateinamens inklusive Pfad und aktuellem Blattnamen  
 "=ZELLE("Dateiname";A1)" ergibt "C:\EXCEL\[STARTTAB.XLS]Tabelle1".

(2) Anzeige nur des Dateinamens und des Blattnamens  
 (mittels Textfunktion "RECHTS" die Namen herausschneiden)  
 "=RECHTS(ZELLE("Dateiname";A1);22)" ..oder..  
 "=RECHTS(ZELLE("Dateiname";A1);LÄNGE(ZELLE("Dateiname";A1))-SUCHEN("[";ZELLE("Dateiname";A1);1)+1)"  
 ergibt "[STARTTAB.XLS]Tabelle1"

(3) Anzeige nur des Laufwerks, in dem die Datei gespeichert ist  
 (mittels Textfunktion "LINKS" die Daten herausschneiden)  
 "=LINKS(ZELLE("Dateiname";A1);3)" ergibt "C:\"

(4) Anzeige nur des Pfades, in dem die Datei gespeichert ist  
 (mittels Textfunktion "LINKS" die Daten herausschneiden)  
 "=LINKS(ZELLE("Dateiname";A1);SUCHEN("[";ZELLE("Dateiname";A1);1)-1)" ergibt "C:\EXCEL\"

Hinweis: Die Funktionsweise der Textfunktion SUCHEN wird unter [SUCHEN](#) näher erläutert.



## Funktion "ISTFEHLER": Abfangen von eventuellen Fehlerwerten

Die Funktion **ISTFEHLER** gibt den Wert WAHR zurück, wenn der der Funktion übergebene Wert einen Fehler enthält.

**Syntax:** ISTFEHLER(Wert)

- Wert: irgendein Zellenwert (z. B. Text, Zahl, Bezug, Name, Wahrheitswert oder auch eine leere Zelle).

**Beispiel 1:** ISTFEHLER(6/0) gibt den Fehlerwert WAHR zurück, da eine Division durch Null den Fehlerwert "#DIV/0!" erzeugt.

**Beispiel 2:** WENN(ISTFEHLER(6/A2);"";6/A2) gibt bspw. dann eine leere Zelle zurück, falls die Zelle A2 eine Null enthält; dagegen eine 3, wenn A2 eine 2 enthält (Erklärung: Die Gänsefüßchen im Dann-Teil der WENN-Funktion erzeugen einen leeren Zelleninhalt).

**Beispiel 3:** WENN(ISTFEHLER(FINDEN("Dollar";A1;1));"";"Treffer!") gibt dann den Text *Treffer!* zurück, wenn die Zelle A1 unter anderem den Text *Dollar* enthält; ansonsten wird ein leerer Zelleninhalt erzeugt. Mit Hilfe dieses Beispiels kann man importierte umfangreiche Datensätze nach bestimmten Texten darin durchsuchen.



## Funktion "VORZEICHEN": Ermittelt das Vorzeichen einer Zahl

Die Funktion **VORZEICHEN** liefert folgende Werte: eine 1, wenn die Zahl positiv ist; eine 0, wenn die Zahl 0 ist; dagegen -1, wenn die Zahl negativ ist.

**Syntax:** VORZEICHEN(Zahl) → Zahl: beliebige reelle Zahl

**Beispiele:**

VORZEICHEN(10) ergibt 1

VORZEICHEN(0) ergibt 0

VORZEICHEN(-0,001) ergibt -1

**Beispiel 2:** Aus Zellbereich A1:A10 sollen getrennt positive und negative Beträge in 2 Formelzellen addiert werden.

Matrixformel für pos. Zahlen: "={SUMME(WENN(VORZEICHEN(A1:A10)=1;A1:A10))}".

Matrixformel für neg. Zahlen: "={SUMME(WENN(VORZEICHEN(A1:A10)=-1;A1:A10))}".

Die Formeln muss man mittels Tastenkombination "Strg-, Hoch- und Returntaste" bestätigen; die geschweiften Klammern kann man nicht selbst eingeben.

Hinweis: Die Funktionsweise von Matrixformeln wird unter [Exkurs: MATRIXFORMELN](#) näher erläutert.



## Funktion "ZÄHLENWENN": Zellen nach vorgegebenen Suchkriterien zählen

Die Funktion **ZÄHLENWENN** ermittelt die Anzahl der (nichtleeren) Zellen eines Bereiches, deren Inhalte mit den Suchkriterien übereinstimmen.

**Syntax:** ZÄHLENWENN(Bereich;Suchkriterien)

- Bereich: zu durchsuchenden Zellbereich.
- Suchkriterien: Kriterien, die bestimmen, welche Zellen gezählt werden sollen (z.B. in Form einer Zahl oder

eines Textes).

**Beispiel 1:** "`=ZÄHLENWENN(A1:B20;"Papier")`" gibt die Anzahl der Zellen zurück, in denen im Zellbereich von A1 bis B20 der Text "Papier" enthalten ist.

**Beispiel 2:** "`=ZÄHLENWENN(A1:B20;"*")`" gibt die Anzahl der Zellen zurück, in denen im Zellbereich von A1 bis B20 irgendein Text enthalten ist (mittels dem Stern als Stellvertreterzeichen).

**Beispiel 3:** "`=ZÄHLENWENN(A1:B20;"<20")-ZÄHLENWENN(A1:B20;"<=10")`" ermittelt, wieviele Werte des Zellbereiches größer 10 und gleichzeitig kleiner 20 sind.

**Beispiel 4:** Zellen mit Doppelseinträgen farbig formatieren.

Im Bereich A1:A50 sollen für alle Werte, die doppelt oder mehr vorkommen, die Zellen-Hintergrundfarbe Rot eingestellt werden. Auch später eingegebene Werte sollen damit erfasst werden.

Vorgehensweise: die Zelle A1 auswählen, per Menü *Format* und Untermenü *Bedingte Formatierung...* beim Bereich "Formel ist" die Option "Bedingung 1" auswählen. Dann rechts daneben die Formel "`=ZÄHLENWENN($A$1:$A$50;A1)>1`" eintragen. Zum Schluss die gewünschte Zellenhintergrundfarbe mittels Befehl *Format...* auswählen und diese Formatierung auf den Zellbereich A2 bis A50 kopieren.

Anmerkung: Siehe hierzu auch Beispiel 4 vom [Exkurs zu MATRIXFORMELN](#).



## 4. Datumsfunktionen

### Berechnung von Datums- und Uhrzeitangaben in Microsoft Excel

Microsoft Excel speichert Datumsangaben als so genannte fortlaufende Zahlen und Uhrzeitangaben als Dezimalbrüche, da diese als Bruchteile eines Tages angesehen werden können (Ein Tag hat daher den seriellen Wert 1, ein Stunde den Wert 1 geteilt durch 24 = 0,04166667).

Datums- und Uhrzeitangaben sind Werte und können daher addiert, subtrahiert und in anderen Berechnungen verwendet werden. Um beispielsweise den Abstand zwischen zwei Datumsangaben zu ermitteln, kann man die Daten voneinander subtrahieren. Bei Zeitangaben ist hierzu das betreffende Tagesdatum mit anzugeben und dann das Zahlenformat "[h]:mm" zu verwenden (dies ist wichtig, da man mit diesem benutzerdefiniertem Format auch über 24 Stunden hinaus Zeitenergebnisse korrekt darstellen kann).

Man kann eine Datums- oder Uhrzeitangabe als fortlaufende Zahl anzeigen, indem man das Zahlenformat der Zelle, die das Datum oder die Uhrzeit enthält, in das Format 'Standard' ändert.

**Beispiel 1** mit Datumsformat "1900" (Standard-Datumsformat in Excel):

1. Januar 1900 = serieller Wert 1

31. Dezember 1999 = serieller Wert 36525

31.12.1999 12:00 = serieller Wert 36525,5 (die Nachkommazahl 5 bedeutet 12:00 Uhr)

**Beispiel 2:** Rechnen mit Uhrzeiten ohne Datumsangabe (innerhalb des 24-Stunden-Bereiches).

Eine Differenz in Stunden und Minuten soll zwischen 2 Uhrzeiten in 2 Zellen gebildet werden und diese in einer weiteren Zelle mit einem Faktor (Stundenlohn) multipliziert werden.

In den Zellen B2 und C2 verwendet man das Zahlenformat "[h]:mm" und schreibt bspw. 7:00 und 15:00 als Uhrzeit hinein. In Zelle D2 (ebenfalls mit o.g. Zahlenformat) zieht man als Formel Zelle C2 von Zelle B2 ab (`=C2-B2`) und erhält 8:00 Stunden. In den Zellen B3 bis D3 verwendet man Zahlenformat "0,00". Zelle B3 erhält die Formel "`=B2*24`", C3 dagegen "`=C2*24`" und D3 erhält "`=D2*24`".

Durch die Multiplikation mit 24 ist die Zeit berechenbar geworden: die Stunden sind in Industriestunden umgewandelt worden (Achtung: eine Rückumwandlung ins Datumsformat ist nicht mehr möglich). D3 erhält dadurch als Wert 8,00. Noch mal multipliziert mit einem zu bestimmenden Lohnfaktor erhält man dann die Arbeitstags-Lohnsumme.

	A	B	C	D	E	F
1		Zeit 1	Zeit 2	Diff.	Faktor	Lohn in DM
2	Zellenformat [h]:mm	7:00	15:00	8:00		
3	Zellenformat 0,00	7,00	15,00	8,00	25,5	204,00
	Formeln in Zelle 3:	=B2*24	=C2*24	=D2*24		=E3*D3

**Beispiel 3:** Rechnen mit Uhrzeiten ohne Datumsangabe über 24 Stunden hinaus.

Eine Zeitdifferenz in Stunden und Minuten soll zwischen 2 Uhrzeiten über die Null-Uhr-Grenze hinaus gebildet werden. Subtrahiert man in Zelle C1 die größere serielle Zahl (z.B. 22.00 Uhr in Zelle A1) von der kleineren (z.B. 5.00 Uhr in Zelle B1), so meldet Excel den Fehler "#####". Excel kann bei seriellen Zahlen, die nur Uhrzeiten und keine Datumsangaben enthalten, mit dieser Formel nicht rechnen: Nach Mitternacht rechnet Excel mit 0:00 Uhr weiter, so dass 5:00 abzüglich 22:00 eine Minuszeit von -17:00 Uhr darstellt, die als Formelfehler ausgewiesen wird.

Man muss eine Formel erstellen, in welcher die Zeit nach 0:00 Uhr zum nächsten Tag zugewiesen wird: Und zwar indem man bei der Zeit nach 0:00 Uhr eine 1 dazu addiert (die serielle Zahl 1 bedeutet genau ein Tag später). In Verwendung des Zeitenformat "[h]:mm" kann man folgende WENN-Funktion verwenden: "**=WENN(B1<A1;1+B1-A1;B1-A1)**". Nur wenn die 24-Stunden-Grenze überschritten wird, wird die 1 bei der Zeiterrechnung dazu addiert, ansonsten wird die Zeit ganz normal errechnet.

Außer in den nachfolgenden Beispielen ist eine praktische Funktion für Darstellung und Berechnung von Datum & Zeit noch im Bereich der Textfunktionen enthalten.

Siehe unter folgendem Link: [Funktion TEXT: Datum und Zeit austricksen](#).



## Funktion "DATUM": Bestimmtes Datum ermitteln

Die Funktion **DATUM** liefert die fortlaufende Zahl des jeweils angegebenen Datums.

**Syntax:** DATUM(Jahr;Monat;Tag)

- Jahr: eine ganze Zahl im Bereich von (Jahr) 1900 bis (Jahr) 9999
- Monat: eine Zahl, die den Monat des jeweiligen Jahres angibt
- Tag: eine Zahl, die den Tag des jeweiligen Monats angibt

**Beispiel:** DATUM(91; 1; 1) liefert den Wert 33239. Dies ist die fortlaufende Zahl, die dem 1. Januar 1991 entspricht.

In Verbindung mit **JAHR**, **MONAT** und **TAG** liefert **DATUM** das Datum nach Ablauf von zu bestimmenden Jahren, Monaten und Tagen (siehe hierzu u.g. Beispiele). Excel umgeht dabei das Problem von Schaltjahren, so dass das zu ermittelnde Datum korrekt errechnet wird.

**Syntax:**

DATUM(JAHR(Anfangsjahr)+Jahreszahl;MONAT(Anfangsmonat)+Monatszähl;TAG(Anfangstag)+Tageszahl)

- Anfangsjahr: das Jahr, von dem in der Berechnung ausgegangen werden soll
- Jahreszahl: die Anzahl an Jahre, die addiert werden soll
- Anfangsmonat; Monatszahl; Anfangstag, Tageszahl: wie oben erwähnt vorgehen

**Beispiel 1:** "**=DATUM(JAHR(C1)+C2;MONAT(C1)+C2;)**" ergibt den 31.03.97, wenn das Anfangsdatum in Zelle C1 der 01.03.96 und die Aufzähl-Zahl in Zelle C2 die 1 ist (es wird 1 Jahr und 1 Monat - bis zum Monatsende - aufgezählt)

**Beispiel 2:** "**=DATUM(JAHR(C1)+C2;MONAT(C1)+C2;TAG(C1))**" ergibt den 01.04.97, wenn das Anfangsdatum in Zelle C1 der 01.03.96 und die Aufzähl-Zahl in Zelle C2 die 1 ist (es wird 1 Jahr und 1 Monat - diesmal bis zum selben Tag wie am Anfangsdatum - aufgezählt)

**Beispiel 3:** "**=DATUM(JAHR(C1)+C2;MONAT(C1)+C2;TAG(C1))**" ergibt den 06.12.96, wenn das Anfangsdatum in Zelle C1 der 01.07.91 und die Aufzähl-Zahl in Zelle C2 die 5 ist

**Beispiel 4:** "**=DATUM(JAHR(C1);MONAT(C1)+C2;TAG(C1)+C2)**" ergibt den 29.02.96, wenn das Anfangsdatum

in Zelle C1 der 01.03.96 und die Aufzähl-Zahl in Zelle C2 die -1 ist

**Beispiel 5:** Differenz in *ganzen Kalendermonaten* zwischen 2 Datumsangaben errechnen.

Die Funktion MONAT in Verbindung mit der Funktion JAHR liefert die Anzahl an vollen Monaten zwischen 2 Datumsangaben. Im Gegensatz zur Funktion DATEDIFF berücksichtigt diese Formel hier nur volle Kalendermonate. Beispielsweise liegen dann zwischen dem 5.5. und dem 15.9. drei Monate (Juni, Juli, August), zwischen dem 1. Mai und dem 15.9. aber 4 Monate (Mai, Juni, Juli, August).

Die Syntax der Formel ersieht man in der Tabelle.

	A	B	C		C
1	Anfangsdatum	Enddatum	Diff.		Differenz
2	12.01.00	10.03.01	13	<Darstellung Formel>	=MONAT(B2)-MONAT(A2)+WENN(MONAT(A2-1)<=>MONAT(A2);1;0)-WENN(MONAT(B2+1)<=>MONAT(B2);0;1)+((JAHR(B2)-JAHR(A2))*12)

**Beispiel Feiertagsberechnung** (ohne Verwendung der Funktion DATUM):

Falls das Datum in Zelle A1 Heiligabend ist, soll in Zelle B2 der Text "X-Mas" stehen.

"=WENN(UND(TAG(A1)=24;MONAT(A1)=12);"X-Mas";"") ergibt den Text "X-Mas", wenn das Datum in Zelle A1 der 24.12. irgendeines Jahres ist; ansonsten ist die Zelle leer dargestellt. Mit den Funktionen TAG und MONAT lassen sich also feststehende Feiertage in Datumzellen abfangen (z.B. 25.12., 1.11., 3.10. etc.).



## Funktion "DATEDIF": Differenz 2er Datumsangaben ermitteln

Die Funktion **DATEDIF** liefert die Anzahl der Zeitintervalle zwischen 2 zu bestimmenden Datumsangaben. Achtung: Die Funktion ist unverständlichlicherweise weder in der Excel-Hilfe noch im Funktionsassistenten enthalten. Man muss die Formel *selbst* in der Eingabeleiste eingeben. Die Funktion erklärt wird lediglich in der VBA-Hilfe im VBA-Editor unter dem Suchbegriff "DateDiff" - beachte hierzu aber die 2 "f" im Funktionsnamen (auch die Argumente sind dort vertauscht - vergleiche hierzu die Syntax).

**Syntax:** DATEDIF(Date1; Date2; "Interval")

- Date1: Das Anfangsdatum
- Date2: das Enddatum
- Intervall: Angabe des zu ermittelnden Intervalls für die Berechnung (wichtig: in Gänsefüßchen setzen). Folgende Möglichkeiten existieren: "y" für Jahr, "m" für Monat, "d" für Tag, "ym" für Monate minus Jahre, "yd" für Tage minus Jahre, "md" für Tage minus Monate minus Jahre.

**Beispiel 1:** Differenz 2er Datumsangaben in Monaten ermitteln.

Beginndatum: 02.12.1994 in Zelle A1, Enddatum: 12.03.1996 in Zelle B1 (macht dann genau 15 Monate Differenz).

Formel: "=DATEDIF(A1;B1;"m")".

Alein unter Verwendung der Funktion MONAT würde man nämlich nur solche Zeiträume ermitteln können, die maximal 12 Monate auseinander liegen.

**Beispiel 2:** Differenz 2er Datumsangaben in Jahren, Monaten und Tagen ermitteln.

Beginndatum: 15.06.1999 in Zelle A1, Enddatum: 14.05.2001 in Zelle B1 (macht dann genau 1 Jahr, 10 Monate und 29 Tage).

Formel: "=DATEDIF(A1;B1;"y")&" Jahre, "&DATEDIF(A1;B1;"ym")&" Monate und "&DATEDIF(A1;B1;"md")&" Tage".

(Zeilenumbruch in der Formelzelle nicht vergessen !)



## Funktion "EDATUM": Fortlaufendes Monatsdatum ermitteln

Die Funktion **EDATUM** liefert die fortlaufende Zahl des Datums, das die angegebene Anzahl an Monaten vor oder hinter dem Ausgangsdatum liegt.

Hinweis: Ist diese Funktion nicht verfügbar, so muss man vorher die *Analyse-Funktionen* über den Befehl *Add-In-Manager* im Menü *Extras* aktivieren.

**Syntax:** EDATUM(Ausgangsdatum; Monate)

- Ausgangsdatum: fortlaufende Datumszahl, die das Ausgangsdatum angibt.
- Monate: die Anzahl an Monaten vor oder nach Ausgangsdatum (ein positiver Wert für Monate ergibt ein zukünftiges Datum, ein negativer Wert ergibt ein zurückliegendes Datum; ist Monate keine ganze Zahl, werden deren Nachkommastellen abgeschnitten)

**Beispiele:**

EDATUM("15.01.91";1) ergibt die serielle Zahl 33284 (gleich dem Monat 15.02.91)

EDATUM("31.03.91";-1) ergibt die serielle Zahl 33297 (gleich dem Monat 28.02.91)

EDATUM("01.01.96";0) ergibt die serielle Zahl 35065 (gleich dem Monat 01.01.96)



## Funktion "MONATSENDE": Den letzten Tag eines Monats ermitteln

Die Funktion **MONATSENDE** liefert die fortlaufende Zahl des letzten Tages des Monats, der um die angegebene Anzahl an Monaten vor oder hinter dem Ausgangsdatum liegt. Mit ihr kann man Rückzahlungs- oder Fälligkeitstermine berechnen, die auf den letzten Tag eines Monats fallen.

Hinweis: Ist diese Funktion nicht verfügbar, so muss man vorher die *Analyse-Funktionen* über den Befehl *Add-In-Manager* im Menü *Extras* aktivieren.

**Syntax:** MONATSENDE(Ausgangsdatum; Monate)

- Ausgangsdatum: fortlaufende Zahl, die dem Anfangsdatum entspricht.
- Monate: gibt an, wie viele Monate vor oder hinter dem Ausgangsdatum liegen sollen (ein pos. Wert für Monate ergibt ein in der Zukunft, ein neg. Wert ergibt ein in der Vergangenheit liegendes Datum; ist Monate keine ganze Zahl, so werden deren Nachkommastellen abgeschnitten)

**Beispiele:**

MONATSENDE("01.01.96"; 1) ergibt die serielle Zahl 35124 (gleich dem Datum 29.02.96)

MONATSENDE("01.01.93"; -1) ergibt die serielle Zahl 33969 (gleich dem Datum 31.12.92)



## Funktion "KALENDERWOCHE": Dieselbige anhand eines Datums ermitteln

Die Funktion **KALENDERWOCHE** liefert eine Zahl, die angibt, in welche Woche des zugehörigen Jahres das anzugebende Datum fällt.

**Syntax:** KALENDERWOCHE(Datum) → Datum: der zu wählende Tag

**Beispiel:** KALENDERWOCHE("15.11.96") ergibt 46 (gleich der 46. Kalenderwoche)



## Funktion "WOCHENTAG": Feststellen, ob Datum in Zelle ein Wochenendtag ist

Mit der Funktion **WOCHENTAG** ermittelt man den Wochentag eines Datums. Dabei liefert die Funktion eine 1, wenn es ein Sonntag ist, eine 2 für einen Montag usw. bis zu einer 7 für den Samstag.

**Syntax:** WOCHENTAG(Zahl;Typ)

- Zahl: eine Kodierung für ein Datum, die Excel für die Datumsberechnung verwendet (Im Beispiel ist eine Zelle angegeben, worin sich ein Datum befindet).
- Typ: eine Zahl, die den Typ des Rückgabewertes bestimmt (optionales Argument)

**Beispiel 1:** In Zelle A1 und A2 wird ermittelt, ob das Datum ein Wochentag ist oder nicht.

	A	B (Formeltext)	B (Ausgabe)
1	27.03.99	=WENN(ODER(WOCHENTAG(A1)=1;WOCHENTAG(A1)=7);"Woch.-Ende";"Arbeitstag")	Woch.-Ende
2	29.03.99	=WENN(ODER(WOCHENTAG(A2)=1;WOCHENTAG(A2)=7);"Woch.-Ende";"Arbeitstag")	Arbeitstag

**Beispiel 2:** Aus einer Kalenderwoche den Montag ermitteln.

In Verbindung mit der Funktion **DATUM** lässt sich der Montag errechnen.

Wenn in Zelle A1 das Jahr (vierstellig!) und in A2 die Kalenderwoche steht, so ermittelt in Zelle A3 die Formel "**=DATUM(A1;1;7\*A2-3-WOCHENTAG(DATUM(A1;;);3))**" das Datum des Montag.



## 5. Textfunktionen

### Funktion "TEXT": Datum und Zeit austricksen

Die Funktion **TEXT** wandelt einen Wert unter Berücksichtigung eines speziellen Zahlenformats in einen Text um. Der jeweilige Wert wird in einen formatierten Text umgewandelt, und das Ergebnis wird nicht länger als Zahl berechnet.

**Syntax:** TEXT(Wert; Textformat)

- Wert: ein numerischer Wert, eine Formel oder ein Bezug auf eine Zelle
- Textformat: ein als Zeichenfolge angegebenes Zahlenformat

**Beispiel 1:** Umbrechen von langen Datumsformaten in Format Text.

Ist in einer Formelzelle - beispielsweise mit der Funktion **HEUTE()** darin - die Spaltenbreite zu klein gewählt, so wird der Zelleninhalt "ausgerautet" anstatt korrekt angezeigt. Dies verhindert man unter Zuhilfenahme der Funktion **TEXT** folgendermaßen: "**=TEXT(HEUTE());"TTTT TT. MMMM JJJJ"**". Man erhält dann zum Beispiel das aktuelle Tagesdatum "Montag 15.März 1999" mit dem Zellenformat *Text*, wobei das Datum über den angrenzenden Zellenrand hinaus sichtbar gemacht wird (vorausgesetzt, diese angrenzenden Zellen sind leer). Das benutzerdefinierte Datumsformat "TTTT TT.MMMM JJJJ" ist dabei in der Funktion selbst mit angegeben.

**Beispiel 2:** Darstellen negativer Zeiten (in Excel normal nicht möglich).

In Verbindung mit der Wenn-Funktion sollen bei Berechnung von Arbeitsstunden, die in den Zellen A1 und A2 dargestellt sind, diese voneinander abgezogen werden. Wird eine negative Zeit errechnet (der Wert in Zelle A2 ist größer als in der in A1), so stellt Excel die Zelle mit Rauten ("###..") als Fehlerwert dar. Folgende Formel verhindert dies:

**"=WENN(A2>A1;TEXT(A2-A1;"- [h]:mm");A1-A2)".** Ist das Ergebnis negativ, so wird es linksbündig in Format *Text* umgewandelt, andernfalls wird ein positives Ergebnis rechtsbündig als Wert geliefert. Daher kann man mit diesen negativen Ergebnissen auch nicht weiterrechnen.

Anmerkung: Es funktioniert auch ohne Verwendung der **TEXT**-Funktion. Mit der Formel "**=WENN(A2-A1<0;(A2-A1)\*-1;A2-A1)"** wird die (negative) Differenz allerdings immer als positiver Wert angezeigt.

**Beispiel 3:** Ein Datum und einen Text zusammen in einer Zelle anzeigen.

In der Zelle A1 steht das aktuelle Tagesdatum, in Zelle B1 der Text "/ Zeichen". Beides soll in einer Zelle zusammen ausgegeben werden.

Die Formel "=TEXT(A1;"TT.MM.JJJJ ")&B1" in Zelle C1 ergibt zum Beispiel "05.05.2000 /Zeichen"



## Funktion "TEIL" : Zeichen aus einer Zeichenfolge ausschneiden

Die Funktion **Teil** liefert eine bestimmte Anzahl Zeichen einer Zeichenfolge ab der von Anwender bestimmten Stelle.

**Syntax:** TEIL(Text; Erstes\_Zeichen; Anzahl\_Zeichen)

- Text: Die Zeichenfolge, die die Zeichen umfasst, die man herauskopieren will
- Erstes\_Zeichen: Die Position des ersten Zeichens, das man aus dem Text herauskopieren will (für das erste Zeichen von Text gilt, dass Erstes\_Zeichen den Wert 1 hat, das zweite den Wert 2, u.s.w.)
- Anzahl\_Zeichen: Gibt an, wie viele der zu "Text" gehörenden Zeichen herauskopiert werden sollen (ist Anzahl\_Zeichen negativ, so liefert die Funktion den Fehlerwert #WERT! )

**Beispiele:**

TEIL("Wonnemonat Mai"; 1; 10) ergibt "Wonnemonat"

TEIL("Wonnemonat Mai"; 12; 3) ergibt "Mai"

**Beispiel 2:** Aus einer Excel-Liste von Telefonnummern will man mittels einer Summenformel aufzählen, wie viele der Nummern aus einer bestimmten Region kommen: bspw. will man wissen, wie viele der 10 Vorwahlen (in dem Bereich A1 bis A10 liegend) mit der Zahlenkombination 026 beginnen. Bezeichnend hierbei ist, dass die Vorwahlen dabei unterschiedliche Zahlenlängen haben dürfen (bspw. 040 oder 02651), da die Funktion TEIL nur die ersten 3 Ziffern berücksichtigt.

Die Array-Formel lautet {=SUMME(WENN(TEIL(A1:A10;1;3)="026";1;0))}.

[die 1 u. 0 am Formelende bedeuten, dass die Funktion SUMME als Array die Spalte A addiert (=1) bzw. nicht addiert (=0)]

Anmerkung: Die Funktionsweise von Matrixformeln wird unter [Exkurs: MATRIXFORMELN](#) näher erläutert.



## Funktion "ANZAHL": Die Anzahl an Zahlen in einem Bereich oder in einzelnen Zellen ermitteln

Mit der Funktion **ANZAHL** ermittelt man, wieviele Zahlenwerte in einem Zellbereich oder in einzelnen Zellen vorhanden sind. Alle anderen Zellinhalte, wie bspw. Formeln oder Text, werden nicht mit aufgezählt.

Hinweis: Mit der Funktion **ANZAHL2** werden dagegen *alle* nichtleeren Zellen aufsummiert.

**Syntax:** ANZAHL(Wert1;Wert2;...)→ Wert1 (etc.): bis zu 30 Zellen oder Zellbereiche können angegeben werden.

**Beispiel 1:** Befinden sich im Zellbereich A1:B10 insgesamt 5 Zahlenwerte, so ermittelt die Formel "=ANZAHL(A1:B10)" den Wert 5.

**Beispiel 2:** Die Funktion ANZAHL in Verbindung mit SUCHEN zählt aus einer Spalte mit diversen Namen (d.h. Zeichenfolgen) diejenigen, die einen identischen Anfang besitzen. Die Formel hierzu muss als Matrixformel/Arrayformel eingegeben werden (Tastenkombination "Strg & Umschalttaste & Return"). Die Funktion ANZAHL ignoriert alle Argumente, die mit einem Text belegt, der nicht in eine Zahl umgewandelt werden kann. Im Beispiel werden die Textzellen aber *aufsummiert*, so dass hier trotzdem die Funktion funktioniert.

	A	B		A	B
1	Name				
2	Müller, Gerd				
3	Meier, Hans				
4	Müller, Paul				
5					
6	Suchname:	Müller	<i>Formeln &gt;</i>	Suchname:	Müller
7	Anzahl:	2	<i>&lt; Darstell.</i>	Anzahl:	{=ANZAHL(SUCHEN(B6;A2:A4))}

Anmerkung:

Die Funktionsweise von Matrixformeln wird unter [Exkurs: MATRIXFORMELN](#) näher erläutert.

Die Funktionsweise der Textfunktion SUCHEN wird unter [SUCHEN](#) näher erläutert.



## Funktion "SUCHEN": Zeichen aus Textzellen herausselektieren

Mit der Funktion **SUCHEN** in Verbindung mit **LÄNGE**, **LINKS** und **RECHTS** lassen sich Zeichen aus Textzellen herausselektieren. Mit den Funktionen LINKS und RECHTS schneidet man die gewünschten Daten heraus; beim zweiten herauszuschneidenden Textteil zieht die Formel dabei das Ergebnis der LÄNGE-Funktion vom SUCHEN-Ergebnis ab.

Im Gegensatz zur Funktion FINDEN unterscheidet SUCHEN nicht zwischen Groß- bzw. Kleinschreibung.

**Syntax:** SUCHEN(Suchtext;Text;Erstes\_Zeichen)

- Suchtext: gesuchtes Zeichen → im u.g. Beispiel das Leerzeichen
- Text: Textzelle, in der man nach dem Zeichen suchen möchte
- Erstes\_Zeichen: die Nr. des Zeichens im Text, ab der man mit der Suche beginnen will (von links nach rechts).

**Beispiel:** Vor- und Nachname einer Person aus einer Zelle in zwei Zellen aufsplitten.

	A	B	C
1	Max Müller	=LINKS(A1;SUCHEN(" ";A1;1)-1)	=RECHTS(A1;LÄNGE(A1)-SUCHEN(" ";A1;1))
2	Wolfgang Wolf	Wolfgang	Wolf



## Funktion "GLÄTTEN": Überflüssige Leerzeichen aus Zelle entfernen

Importiert man Datensätze aus einer Textverarbeitung oder aus einer Datenbank, so sind möglicherweise überflüssige Leerzeichen enthalten. Für eine solche Situation steht die Funktion **GLÄTTEN** zu Verfügung. Als Argument benötigt sie lediglich den Text, aus dem die Leerzeichen entfernt werden sollen. Diese können direkt oder als Zellbezug angegeben werden (d.h. bspw. in den Zellen rechts der Textzellen die Funktion erstellen, dabei beziehen auf linke Zellen, dann nach erfolgter Glättung die rechten Zellen auf die linken verschieben / überschreiben).

- als überflüssig werden alle Leerzeichen angesehen, die am Anfang oder am Ende stehen
- die Anzahl der Leerzeichen zwischen 2 Wörtern wird auf eins reduziert, falls sich dort mehrere befinden sollten

**Syntax:** GLÄTTEN(Text) → Text: der Text, aus dem man überflüssige Leerzeichen entfernen möchte

**Beispiel:** GLÄTTEN(" Gewinn im ersten Quartal ") ergibt "Gewinn im ersten Quartal".



## Funktion "VERKETTEN": Zusammenfassen von Zellinhalten

Die Funktion **VERKETTEN** kann Informationen von 2 oder mehreren Zellen in einer einzigen Zelle zusammenfassen als auch Text in einer Zelle mit einer Formel verbinden.

Hinweis: Anstelle der Funktion kann ebenso der Operator "&" verwendet werden, um als Zeichenfolgen vorliegende Elemente miteinander zu verbinden.

**Syntax:** VERKETTEN (Text1;Text2; ...) → Text1; Text2; ... sind 1 bis 30 Argumente, die Zeichenfolgen angeben, die zu einer Zeichenfolge verbunden werden sollen. Als Argumente dürfen Zeichenfolgen (Texte), Zahlen und Bezüge auf einzelne Zellen verwendet werden.

**Beispiel 1:** Verketten des Vornamens aus Zelle A1 mit dem Familiennamen aus Zelle B1: in Zelle C1 die Formel "`=VERKETTEN(A1;" ";B1)`" eingeben [→ bei 2. Argument Leerzeichen zwischen Gänsefüßchen, um Abstand zwischen den Worten zu halten !]

**Beispiel 2:** Verketten des Textes "Rechnungsbetrag brutto: DM" mit der Summe aus den Zellen A2:B2 in Zelle C2: in Zelle C2 die Formel "`=VERKETTEN("Rechnungsbetrag brutto: DM ";SUMME(A2:B2))`" eingeben

	A	B	C
1	Al	Bundi	Al Bundi
2	25	25	Rechnungsbetrag brutto: DM 50



## Funktion ERSETZEN: Bestimmtes Zeichen eines Zelleninhaltes löschen

Die Funktion **ERSETZEN** ersetzt eine bestimmte Anzahl Zeichen ab einer bestimmten Stelle innerhalb eines Textes.

**Syntax:** ERSETZEN(Alter\_Text;Erstes\_Zeichen;Anzahl\_Zeichen;Neuer\_Text)

- Alter\_Text: der Text, in dem man einige Zeichen ersetzen möchte.
- Erstes\_Zeichen: die Position des zu Alter\_Text gehörenden Zeichens, an der mit dem Ersetzen durch Neuer\_Text begonnen werden soll.
- Anzahl\_Zeichen: gibt an, wie viele der zu Alter\_Text gehörenden Zeichen man durch Neuer\_Text ersetzen möchte.
- Neuer\_Text: der Text, durch den die zu Alter\_Text gehörenden Zeichen ersetzt werden.

**Beispiel 1:** ERSETZEN(A1;1;1; "") löscht das erste Zeichen in der (Text-)Zelle A1 (egal, wie lang die Zeichenfolge auch ist).

**Beispiel 2:** ERSETZEN(A1;LÄNGE(A1); 1; "") löscht in Zusammenarbeit mit der Funktion LÄNGE das letzte Zeichen der Zelle A1. Mit der Funktion LÄNGE wird die Länge der Zeichenkette bestimmt: gibt man für Anzahl\_Zeichen die 1 an, so wird genau das letzte Zeichen eliminiert.



## Funktion "WECHSELN": Zeichen innerhalb einer Zelle versetzen

Die Funktion **WECHSELN** tauscht einen alten Text durch einen neuen Text in einer Zeichenfolge aus. Im Beispiel werden damit aber Zahlen bearbeitet.

**Syntax:** WECHSELN(Text;Alter\_Text;Neuer\_Text;ntes\_Auftreten)

- Text: der Text (oder der Bezug auf eine Zelle, die den Text enthält), in dem man Zeichen ersetzen möchte
- Alter\_Text: die Zeichenfolge, die man wechseln möchte
- Neuer\_Text: die Zeichenfolge, durch die man Alter\_Text ersetzen möchte
- ntes\_Auftreten: gibt an, das wievielte Vorkommen von Alter\_Text man durch Neuer\_Text ersetzen möchte. Wenn man ntes\_Auftreten angibt, wird nur dieses Vorkommen von Alter\_Text ersetzt; andernfalls wird Alter\_Text an jeder Stelle, an der er in Text vorkommt, durch Neuer\_Text ersetzt.

**Beispiel:** SAP-Habenbeträge als normale Minuszahlen konvertieren.

Haben-Bilanzzahlen in der Buchungssoftware SAP besitzen ein nachgestelltes Minus-Zeichen rechts der Ziffer als Kennung. Um mit Soll- und Haben-Zahlen rechnen zu können, verwendet man für einzelne Zellen folgende Formel (dann wird aus "12-" nämlich "-12" werden, falls Minuszahlen vorhanden sind; ist dagegen die Zahl 12 ohne Minuszeichen, so wird sie eben nochmal als 12 dargestellt):  
`=WENN(RECHTS(A1;1)="-";("-"&WECHSELN(A1;"-";""))*1;A1)`



## Funktion "WIEDERHOLEN": wiederholt einen Text so oft wie angegeben.

**Syntax:** WIEDERHOLEN(Text;Multiplikator)

- Text: der Text, in dem man bestimmte Zeichen einsetzen möchte.
- Multiplikator: gibt an, wie oft das Trennzeichen eingefügt werden soll (positive Zahl).

**Beispiel 1:**

"WIEDERHOLEN("1+"; 3)" ergibt "1+1+1+".

**Beispiel 2:** Ein einfaches Histogramm erzeugen.

Ganzzahlen einer Wertetabelle lassen sich als Balkendiagramm darstellen: ist in Zelle A1 der Wert 10 und in Zelle A2 die Funktion "`=WIEDERHOLEN(Zeichen(0129);A1)`" so ergibt sich folgende Ausgabe: "□□□□□□□□□□". Der Platz ist natürlich begrenzt wegen der Zeichengröße und deren Anzahl. Wenn man die Ausrichtung der Funktionszelle senkrecht stellt, erhält man ein Säulendiagramm.

**Beispiel 3:** Trennzeichen 'Bindestrich' zwischen Ziffern erstellen.

Steht in Zelle A2 der Wert "12631" und in B2 die Funktion "`=TEIL(TEXT(A2;WIEDERHOLEN("0-";LÄNGE(A2)));1;LÄNGE(A2)*2-1)`"

so ergibt sich folgende Ausgabe in Zelle B2: "1-2-6-3-1". Normaler Text wird nicht umgesetzt. Will man Nullen vor den Zahlen berücksichtigen, so muss man dem Wert in Zelle A2 ein Hochkomma voranstellen.



## Exkurs: Matrix- bzw. Array-Formeln verwenden

**Matrixformeln** können - im Gegensatz zu normalen Formeln - mehrere Berechnungen durchführen und dann entweder ein einzelnes Ergebnis oder mehrere Ergebnisse liefern. Sie können eine ganze Gruppe von Zellen gleichzeitig bearbeiten: Anstatt zeitraubend mehrere unterschiedlich zu definierende Einzelformeln zu erstellen, wird eine einzige Matrixformel verwendet (siehe Beispiele). Anders ausgedrückt: aus einer Gesamtmenge wird ein Teil der Daten daraus ausgewertet mittels vorzugebenden Bedingungen. Statistische Auswertungen sind die häufigsten Anwendungsgebiete für Matrixformeln.

Matrixformeln bearbeiten zwei oder mehr Wertesätze, die als **Matrixargumente** bezeichnet werden. Alle Matrixargumente müssen über dieselbe Anzahl von Zeilen und Spalten verfügen. Um mit Hilfe einer Matrixformel mehrere Ergebnisse zu berechnen, muss man die Matrix in einen Zellbereich eingeben, der über dieselbe Anzahl

von Zeilen und Spalten verfügt wie die Matrixargumente.

Es ist zu beachten, dass maximal 4096 Elemente einer Matrix ansprechbar sind, ansonsten kommt es zu einer Fehlermeldung.

Eine Matrixformel muss immer in geschweifte Klammern "{}" eingeschlossen werden. Die Klammern werden durch die Tastenkombination "STRG+Umschalttaste+Return" erstellt (vorher muss man aber in der Excel-Bearbeitungsleiste die Formel anklicken, so dass der Cursor als Strich in der Formel erscheint). Excel fügt daraufhin der Formel die Klammern hinzu.

Excel enthält eine Reihe von bereits vorgefertigten Matrixfunktionen, die meisten findet man im Funktions-Assistenten unter der Kategorie "Matrix". Folgende dieser so genannten *integrierten* Matrixformeln sind in dieser Homepage unter Kategorie 'Logische Funktionen' dargestellt: [BEREICH.VERSCHIEBEN](#), [INDEX](#), [INDIREKT](#), [SVERWEIS](#), [VERGLEICH](#), [WAHL](#) und [ZEILE](#).

Hinweis: Logische Funktionen wie bspw. UND oder ODER lassen sich als Matrixformeln nicht einsetzen.

**Beispiel 1:**  $\{=MAX(LÄNGE(A1:A100))\}$  ermittelt die Zelle aus dem Zellenbereich A1 bis A100 mit dem längsten Inhalt/Zeichenfolge (Text oder Zahl).

**Beispiel 2:**  $\{=SUMME((A1:A100="X")*(B1:B100="Y")*(C1:C100))\}$  addiert aus dem Zellenbereich C1:C100 die Zellen (mit Zahlenwerten darin), wenn in A1:A100 ein 'X' und in B1:B100 ein 'Y' eingetragen wurde.

**Beispiel 3:** Berechnung eines Jahreszinsbetrages mit verschiedenen Anlagebeträgen und verschiedenen Zinssätzen (bei höherem Anlagebetrag gibt es auch höhere Zinsen) mittels folgender Matrixformel:  $\{=A2:A4*\{6;6.5;7\}/100\}$  in den Zellen B2 bis B4. Hinweis: Die inneren geschweiften Klammern der Formel werden im Gegensatz zu den äußeren Klammern per Tastenkombination "ALT+123" und "ALT+125" erstellt.

	A	B		A	B
1	Anlage	Zinsbetrag		Anlage	Zinsbetrag
2	2.000	120	Formeln >	2.000	$\{=A2:A4*\{6;6.5;7\}/100\}$
3	3.000	195	< Darstell.	3.000	$\{=A2:A4*\{6;6.5;7\}/100\}$
4	4.000	280		4.000	$\{=A2:A4*\{6;6.5;7\}/100\}$

Im vorgenannten Beispiel bekommt man bei 2 Tsd. EURO 6%, bei 3 Tsd. EURO 6,5% und bei 4 Tsd. EURO 7% Zinsen.

**Beispiel 4:** In einer Liste im Zellbereich A2 bis A30 die mehrfach vorkommenden Zellinhalte herausfinden. In Zelle B3 folgende Matrixformel eintragen:  $\{=WENN(SUMME(WENN(A3=$A$2:A2;1;0))>0;"doppelt";"")\}$  und mittels Kopieren & Einfügen in die restlichen Zellen von B4 bis B30 übertragen (die Zelle B2 ist ohne Formel, da ein Wert dort sowieso zum ersten Mal erscheint).

Kommen Zellinhalte in Spalte A mehrfach vor, so wird rechts daneben in Spalte B der Text "doppelt" erscheinen.

Anmerkung: Siehe hierzu auch Beispiel 4 von der Funktion [ZÄHLENWENN](#).

**Beispiel ohne Arrayformel** (d.h. ohne geschweifte Klammern verwenden zu müssen): Personalkosten pro Arbeitstag errechnen, wobei ein bis drei Arbeitsschichten anfallen können. Für die Tabelle existiert nur eine einzige Formel, die auf alle gewünschten Zellen kopierbar ist.

	A	B	C	D	Spalte B bis D
1	DM	Schichten-anzahl	Schichten-anzahl	Schichten-anzahl	Formeln (⇌ links normale Darstellungen)
2		1	2	3	
3	150	150	300	450	$\{=($A$3:$A$5)*($B$2:$D$2)\}$
4	155	155	310	465	$\{=($A$3:$A$5)*($B$2:$D$2)\}$
5	160	160	320	480	$\{=($A$3:$A$5)*($B$2:$D$2)\}$



## Exkurs2: Gültigkeitsprüfungen mit Excel-Funktionen

Mit **Gültigkeitsprüfungen** kann man erreichen, dass in einer Tabelle vom Anwender nur bestimmte, gültige Daten

eingetragen werden dürfen. Aufrufen kann man die Prüfung mittels Menü *Daten*, Untermenü *Gültigkeit*. Im Feld *Zulassen* verwendet man die Option *Benutzerdefiniert*, um im Feld *Formel*: die gewünschten Formelfunktionen eingetragen zu können. Das Häkchen beim Optionsfeld *Leere Zellen ignorieren* muss hierzu unbedingt bestehen bleiben (Dieses Kontrollkästchen betrifft nur die Frage, ob eine leere Eingabezelle - die mit der Gültigkeitsprüfung ausgestattet ist - als "gültig" gewertet werden soll oder nicht).

Im Register *Fehlermeldung* kann man bei den Feldern *Stil*, *Titel* & *Fehlermeldung* die entsprechenden Angaben machen.

Soll die Prüfung für andere Zellen auch gelten, so markiert man diese inklusive der bisherigen Prüfzelle und ruft nochmal das Menü *Daten*, Untermenü *Gültigkeit* auf. In einem Dialogfenster wird man gefragt, ob die Prüfung nun auf die gesamte Markierung erweitert werden soll.

**Beispiel 1 (noch ohne Excel-Funktion):** In Zelle A1 die Eingabewerte auf ganze Zahlen zwischen 0 bis 10 beschränken.

In Feld *Zulassen* aktiviert man hierzu die Option "Ganze Zahl". Für das nun aktive Feld *Daten* setzt man zwischen *Minimum*: "0" und *Maximum*: die Zahl "10" ein.

**Beispiel 2:** In Spalte A nur Zahlen als Eingabe zulassen.

Die Zelle A1 selektieren, dann in der Gültigkeitsprüfung unter Feld *Zulassen* die Option "Benutzerdefiniert" aktivieren. Die Formel "=ISTZAHL(A1)" eintragen. Dann das Zellenformat von Zelle A1 mittels Menü *Bearbeiten* - *Inhalte Einfügen* - *Option Gültigkeit* auf die restlichen Zellen in Spalte A kopieren.

Mittels "=ISTTEXT(A1)" wird die Eingabe auf Textinhalte beschränkt.

**Beispiel 3:** Doppelte Listeinträge verhindern.

In Spalte A sollen doppelte Einträge vermieden werden. Die Zelle A1 selektieren, dann in der Gültigkeitsprüfung unter Feld *Zulassen* die Option "Benutzerdefiniert" aktivieren. Die Formel "=ZÄHLENWENN(A:A;A1)=1" eintragen. Dann das Zellenformat von Zelle A1 mittels Menü *Bearbeiten* - *Inhalte Einfügen* - *Option Gültigkeit* auf die restlichen Zellen in Spalte A kopieren.

Siehe hierzu auch die Beschreibung zur Funktion [ZÄHLENWENN](#).

**Beispiel 4:** Warnung vor einer Änderung des Zelleninhaltes.

Die Zellen A1 bis A5 sollen vor Änderungen geschützt werden, ohne dass der Excel-Blattschutz hierzu aktiviert werden muss. Dies ist allerdings nur als Warnung vor Änderungen an der Zelle zu verstehen - der Zelleninhalt kann nämlich trotzdem vom Anwender mittels Entf-Taste gelöscht werden:

Zelle A1 selektieren, dann in der Gültigkeitsprüfung unter Feld *Zulassen* die Option "Benutzerdefiniert" aktivieren. Die Formel "=ISTLEER(INDIREKT("ZS";0))" eintragen. Im Register *Fehlermeldung* die Meldung schreiben, die vor einer Zellenänderung warnen soll. Dann das Zellenformat von Zelle A1 mittels Menü *Bearbeiten* - *Inhalte Einfügen* - *Option Gültigkeit* auf die restlichen Zellen in Spalte A kopieren.

Siehe hierzu auch die Beschreibung zur Funktion [INDIREKT](#).

---

[Link zur Homepage-Einstiegsseite](#)